HRP 262 SAS LAB EIGHT, May 28, 2008


Lab Eight: Mixed Models and GEE; time-dependent predictors; modeling change

Lab Objectives

After today’s lab you should be able to:

1. Plot individual and mean profile plots where both the predictor and the outcome are changing over time.

2. Analyze longitudinal data with mixed models using PROC MIXED.

3. Analyze longitudinal data with GEE using PROC GENMOD.

4. Interpret results from mixed models and GEE.

5. Understand the difference between “between-subjects” and “within-subjects” effects.

6. Consider different ways to probe within-subjects effects for time-dependent predictors.

 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. This week we’re going to use the data I presented in class last Monday. Goto the class website (www.stanford.edu/~kcobb/courses/hrp262) and cut and paste these data (under lab 8 data).

data broad;

input id    treat  time1   time2   time3   time4  chem1 chem2 chem3 chem4 ;

datalines; 

1   1    20    18       15       15      1000   1100  1200  1300

2   1    22    24       18       22      1000   1000  1055  950

3   1    14    10       24       10      1000   1999  800   1700

4   1    38    34       32       24      1000   1050  .     1400

5   1    25    29       25       29      1000   1020  1040  1045

6   1    30    28       26       14      1000   1100  1180  1500

7   0    20    15       21       20      1000   1200  1200  1000

8   0    21    27       19       27      1000   900   1075   900

9   0    15    22       22       20      1000   .     1000   700

10  0    39    .        39       34      1000   950   1033   1025

11  0    27    27       31       22      1000   950   910    1050

12  0    28    24       33       .       1000   1015  985   .

;

run;
2. Turn these data into the long form.
data long;

set broad;

time=0; score=time1; chem=chem1; output;

time=2; score=time2; chem=chem2; output;

time=3; score=time3; chem=chem3; output;

time=6; score=time4; chem=chem4; output;

label time='Months';

label chem='chemical levels';

label score='depression score';

run;

3. Ignoring treatment group to begin with, graph the development of chemical over time against the development of depression score over time as follows.


goptions reset=all ftext=titalic htext=2;

axis1 label=(angle=90);
axis2 label=(angle=90 c=red);

symbol1 i=join l=1 c=black  r=12;

symbol2 i=join  l=2 c=red r=12;

proc gplot data=long;

plot score*time=id / nolegend vaxis=axis1;

plot2 chem*time=id/ vaxis=axis2 nolegend;

run; quit;

4. Graph profile plots by individual:

symbol1 i=join l=1 c=black v=none; 

symbol2 i=join  l=2 c=red v=none;

proc gplot data=long;

plot score*time / nolegend vaxis=axis1;

plot2 chem*time/ vaxis=axis2 nolegend;

by id;

run; quit;

5. Graph the mean lines as follows (with a 1-standard deviation bar):


goptions reset=symbol; 

symbol1 i=stdm1j l=1 c=black v=none; 

symbol2 i=stdm1j  l=2 c=red v=none;

proc gplot data=long;

plot score*time / nolegend vaxis=axis1;

plot2 chem*time/ vaxis=axis2 nolegend;

run; quit;

6. Also graph the means by treatment group:


proc sort data=long;

by treat;

run;
symbol1 i=stdm1j c=black l=1 v=none;

symbol2 i=stdm1j  c=red l=1 v=none;

proc gplot data=long;

plot score*time /vaxis=axis1 nolegend;

plot2 chem*time / vaxis=axis2 nolegend;

by treat;

run; quit;

7. And graph depression scores and chemical levels separately by treatment group.

symbol1 i=stdm1jt c=black l=1 v=none;

symbol2 i=stdm1jt  c=red l=1 v=none;
proc gplot data=long;

plot score*time=treat chem*time=treat /vaxis=axis1 ;

run; quit;

8. Practice repeated-measures ANOVA to look at the relationship between treatment and depression: (1) Is there a change in depression scores over time (within-subject effect of time)? (2) Is there a difference in the marginal means of the groups in depression scores (between-subject effect of group)? (3) Do the two groups differ in their response profiles for depression scores (within-subject effect of time*group)?  We have to fill in the data with LOCF first (otherwise we’ll lose two observations, which we can’t afford with only 12 total!)

data broad;

set broad;

if time2=. then time2=time1;

if time3=. then time3=time2;

if time4=. then time4=time3;

run;

proc glm data=broad;

class treat id;

model time1-time4= treat/nouni;

repeated time;

run;

    The GLM Procedure

                             Repeated Measures Analysis of Variance

                        Tests of Hypotheses for Between Subjects Effects

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      treat                        1      130.020833      130.020833       0.72    0.4175

      Error                       10     1818.208333      181.820833

    The GLM Procedure

                             Repeated Measures Analysis of Variance

                   Univariate Tests of Hypotheses for Within Subject Effects

                                                                                    Adj Pr > F

Source                     DF    Type III SS    Mean Square   F Value   Pr > F    G - G    H - F

time                        3     60.3958333     20.1319444      1.15   0.3454   0.3411   0.3454

time*treat                  3     79.2291667     26.4097222      1.51   0.2327   0.2415   0.2327

Error(time)                30    525.6250000     17.5208333


                              Greenhouse-Geisser Epsilon    0.7836

                              Huynh-Feldt Epsilon           1.1421
9. Repeat step (8) with chemical levels as the outcome variable (i.e., does treatment affect chemical level?).

data broad;

set broad;

if chem2=. then chem2=chem1;

if chem3=. then chem3=chem2;

if chem4=. then chem4=chem3;

run;

proc glm data=broad;

class treat id;

model chem1-chem4= treat/nouni;

repeated time;

run;

                                       The GLM Procedure

                             Repeated Measures Analysis of Variance

                        Tests of Hypotheses for Between Subjects Effects

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      treat                        1     271652.5208     271652.5208       6.32    0.0307

      Error                       10     429782.7083      42978.2708


                                       The GLM Procedure

                             Repeated Measures Analysis of Variance

                   Univariate Tests of Hypotheses for Within Subject Effects

                                                                                    Adj Pr > F

Source                     DF    Type III SS    Mean Square   F Value   Pr > F    G - G    H - F

time                        3     125941.063      41980.354      1.25   0.3094   0.3040   0.3082

time*treat                  3     276899.563      92299.854      2.75   0.0602   0.1040   0.0882

Error(time)                30    1008108.125      33603.604


                              Greenhouse-Geisser Epsilon    0.5243

                          Huynh-Feldt Epsilon           0.6672
10. Now, use a mixed model to look at the change in depression score over time, ignoring group and chemical for the moment. Compare a model with a random intercept to the model with random intercept and random time:


proc mixed data=long;

class  id; 

model score= time /solution; 

random int / subject=id ;

run; quit;

proc mixed data=long;

class  id;  

model score=time /solution; 

random int time/ subject=id ;

run; quit;

OUTPUT from random intercept only model: 





Covariance Parameter Estimates

                               Cov Parm      Subject    Estimate

                               Intercept     id          36.7740

                               Residual                  18.0554

                                        Fit Statistics

                             -2 Res Log Likelihood           285.2

                             AIC (smaller is better)         289.2

                             AICC (smaller is better)        289.5

                             BIC (smaller is better)         290.2

                                   Solution for Fixed Effects


                                         Standard

                Effect       Estimate       Error      DF    t Value    Pr > |t|

                Intercept     25.5589      2.0193      11      12.66      <.0001

                time          -0.5016      0.2928      33      -1.71      0.0960                                                          

OUTPUT from random intercept and random time model:
                                      The Mixed Procedure

                                Covariance Parameter Estimates


                               Cov Parm      Subject    Estimate

                               Intercept     id          37.4767

                               time          id          0.08512

                               Residual                  17.4021

                                        Fit Statistics


                             -2 Res Log Likelihood           285.2

                             AIC (smaller is better)         291.2

                             AICC (smaller is better)        291.8

                             BIC (smaller is better)         292.7

                                   Solution for Fixed Effects

                                         Standard

                Effect       Estimate       Error      DF    t Value    Pr > |t|

                Intercept     25.5558      2.0249      11      12.62      <.0001

                time          -0.4981      0.3001      11      -1.66      0.1252
11. Just for fun, try adding a quadratic term for time in the model. Since time is continuous, we are allowed to try response profiles that are non-linear.

proc mixed data=long;

class  id; 

model score= time time*time/solution; 

random int / subject=id ;

run; quit;
                                   Solution for Fixed Effects

                                         Standard

                Effect       Estimate       Error      DF    t Value    Pr > |t|

                Intercept     24.7801      2.1324      11      11.62      <.0001

                time           0.5427      0.9419      32       0.58      0.5685

                time*time     -0.1709      0.1466      32      -1.17      0.2524
12. Next run the mixed model testing for an effect of treatment group on depression scores (time-independent predictor, comparable to the model we ran in rANOVA). Use a random intercept only:

proc mixed data=long;

class  id treat; *treat time as continuous;
model score=  treat time treat*time /solution; 

random int / subject=id ;

run; quit;

Fit Statistics:                AIC (smaller is better)         280.1
  Solution for Fixed Effects

                                              Standard

           Effect        treat    Estimate       Error      DF    t Value    Pr > |t|

           Intercept               25.4756      2.8844      10       8.83      <.0001

           treat         0         0.05830      4.0935      32       0.01      0.9887

           treat         1               0           .       .        .         .

           time                    -0.9911      0.3861      32      -2.57      0.0151

           time*treat    0          1.0559      0.5657      32       1.87      0.0712
           time*treat    1               0           .       .        .         .


13. Run the comparable model in GEE (with an exchangeable correlation matrix), to see that it gives very similar results (identical point estimates, slightly different standard errors):

proc genmod data=long;

class  id treat; 

model score=  treat time treat*time; 

repeated subject = id / type=exch;

run; quit;

The GENMOD Procedure

                              Analysis Of GEE Parameter Estimates

                              Empirical Standard Error Estimates

                                     Standard   95% Confidence

               Parameter    Estimate    Error       Limits            Z Pr > |Z|

               Intercept     25.4756   3.1614  19.2793  31.6718    8.06   <.0001

               treat      0   0.0283   4.3786  -8.5535   8.6101    0.01   0.9948

               treat      1   0.0000   0.0000   0.0000   0.0000     .      .

               time          -0.9911   0.4815  -1.9349  -0.0473   -2.06   0.0396

               time*treat 0   1.0557   0.5585  -0.0389   2.1504    1.89   0.0587
               time*treat 1   0.0000   0.0000   0.0000   0.0000     .      .
14. Next, test for the effect of chemical level, as a time-dependent predictor. 

proc mixed data=long;

class  id ; 

model score=  chem time /solution; 

random int / subject=id ;

run; quit;
  Solution for Fixed Effects

                                         Standard

                Effect       Estimate       Error      DF    t Value    Pr > |t|

                Intercept     40.2416      3.4099      11      11.80      <.0001

                chem         -0.01450    0.002806      30      -5.17      <.0001

            time          -0.1960      0.2329      30      -0.84      0.4067


14. To tease out the within-subjects effect of chemical, try correlating the change in chemical level with the change in depression score:

data change;

set broad;

ctime=2; cscore=time2-time1; cchem=chem2-chem1; output;

ctime=1; cscore=time3-time2; cchem=chem3-chem2; output;

ctime=3; cscore=time4-time3; cchem=chem4-chem3; output;

run;

proc gplot data=change;

plot cscore*cchem cscore*ctime cchem*ctime;

symbol1 v=dot i=rl;

run;

proc mixed data=change;

class  id; 

model cscore= cchem ctime /solution; 

run; quit;

                                      The Mixed Procedure

                                   Solution for Fixed Effects

                                         Standard

                Effect       Estimate       Error      DF    t Value    Pr > |t|

                Intercept      1.6611      2.2163      28       0.75      0.4598

                cchem        -0.01149    0.002429      28      -4.73      <.0001

                ctime         -1.0265      1.0363      28      -0.99      0.3304
15. Finally, try to tease out whether treatment is working through influencing chemical levels.

proc mixed data=long;

class  id; 

model chem= treat time time*treat/solution; 

random int / subject=id ;

run; quit;
   



Solution for Fixed Effects

                                         Standard

               Effect        Estimate       Error      DF    t Value    Pr > |t|


               Intercept      1021.08     67.2967      10      15.17      <.0001

               treat          -0.9694     94.3635      31      -0.01      0.9919

               time          -10.1075     18.9552      31      -0.53      0.5977

               treat*time     57.3642     26.0870      31       2.20      0.0355

proc mixed data=long;

class  id; 

model score= treat time time*treat chem/solution; 

random int / subject=id ;

run; quit;
                                  Solution for Fixed Effects

                                         Standard

               Effect        Estimate       Error      DF    t Value    Pr > |t|

               Intercept      39.4998      4.1810      10       9.45      <.0001

               treat          0.01141      3.9010      29       0.00      0.9977

               time          -0.05807      0.3364      29      -0.17      0.8642

               treat*time     -0.2835      0.4903      29      -0.58      0.5676

               chem          -0.01380    0.003066      29      -4.50      0.0001
DUE FROM LAB 8: Nothing due from lab 8!  Start work on your final exam!
Colors label of second vertical axis red.











AIC is bigger—so prefer the model with only a random  intercept. 





ctime indicates how much time passed between their measurements, in months.





Treatment affects chemical levels over time (as we saw in rANOVA).





Overall, treatment and control groups differ in chemical level.





Beyond treatment’s effect on chemical level, there are no additional treatment effects. 








There’s an overall decrease in depression score over time, and a borderline significant difference between the groups in changes over time. (1 point higher change per month in the treat=0 group). By modeling time as continuous we’re able to pick up more than repeated measures ANOVA.





There’s a borderline significant difference in the change in chemical levels over time between treatment and control.





Random time does not explain much of the variance. 





No evidence of quadratic shape over time.





This model is just OLS regression (I could have used PROC REG).





Shows that within-subjects effects are really important.





AIC=289.2 











Data are already sorted by id. Otherwise, we would need to use proc sort.





The plot2 option adds a second vertical axis. There must be a shared horizontal axis, or this won’t work!





i=stdm1j plots and joins the means at each time with a 1-standard deviation bar. You must select v=none, or SAS will plot the individual points (unjoined) as well.





i=stdm1jt adds whiskers to the std dev bars.





i=std1m1j gives means with a 1-standard error bar





i=std1m1jt  adds whiskers to the standard error bars





Tells SAS to give you the parameter estimates!  Otherwise, they do not appear in the output!!





We need 12 black lines and 12 red lines.








You must sort the data by treatment group before this graph will work.








Rate of change in depression score overall is -.5016 per month. 





Note the inverse relationship between depression score and chem.





Average baseline depression score is 25.6. 






































Chemical is highly correlated to depression score. This is likely due in large part to huge within-subject effects. Since everyone had the same baseline chemical level (1000), between-subjects effects are likely less important. 





























If you omit the overlay option, SAS gives you two separate graphs.





No effects are significant.





Sphericity is not too bad; for GEE, exchangeable correlation structure might be OK.

